

Лабораторная работа №4

РАСЧЕТ НАПРЯЖЕННО-ДЕФОРМИРОВАННОГО СОСТОЯНИЯ, СТАЦИОНАРНОГО РАСПРОСТРАНЕНИЯ ТЕПЛА и СВЯЗАННОЙ ЗАДАЧИ ТЕРМОУПРУГОСТИ В ТРЕХМЕРНОЙ КОНСТРУКЦИИ

КЛЮЧЕВЫЕ СЛОВА

1. Задача теории упругости (структурный анализ).
2. Задача теплопроводности (температурный анализ)
3. Задача термоупругости
4. Статическая задача
5. Пространственная задача
6. Построение 3D-геометрии: вращение, параллельный перенос, экструзия.

ОПИСАНИЕ ЗАДАЧИ

Рассмотрим трехмерную фигуру (рис. 2), полученную из плоской твердотельной модели (рис. 1) путем вращения плоской области вокруг оси.




Рис. 1. Схема области




Рис. 2. Трехмерная конструкция, полученная с помощью вращения

Условие задачи. Дана двумерная область в виде латинской буквы «R». Геометрия области, отнесенной к декартовой системе координат Oxy , показана на рис. 1. Размеры буквы определяются следующими параметрами: $HL=1.4$ м (высота); $WL=0.8$ м (ширина); $WWL=0.2$ м (ширина стенок); $HAL=0.7$ м (дополнительный размер по высоте). Геометрия трехмерной фигуры создается из плоской области в форме буквы «R» путем вращения на 45° вокруг вертикальной оси, отстоящей влево от буквы на расстояние WWL . Примем, что материалом области является сталь с коэффициентом теплопроводности $k=K_{xx}=46.7$ (Вт/(м град)), модулем Юнга $E=2 \cdot 10^{11}$

N/m^2 и коэффициентом Пуассона $\nu=0.29$ и коэффициентом теплового расширения $\alpha=1.51 \times 10^{-5}$ 1/град.

Требуется решить по отдельности стационарную задачу теории упругости (структурный анализ), теплопроводности (температурный анализ) и связанную задачу термоупругости при заданных граничных условиях, а также сравнить результаты, полученные на произвольных тетраэдральных и канонических гексаэдральных сетках.

Структурный анализ. Предположим, что нижние грани буквы жестко закреплены, а верхняя грань растягивается распределенной нагрузкой величины $p=10$ ГПа ($=10 \cdot 10^9$ Н/м²).

Температурный анализ. Предположим, что на левой нижней грани температура θ_h равна 90°C ; а на правой нижней грани температура θ_h равна 180°C . На верхней грани имеем условие конвективного теплообмена: $\mathbf{n} \cdot \mathbf{q} = -h_f(\theta_{ext} - \theta)$, где \mathbf{n} – внешняя единичная нормаль к границе; $\mathbf{q} = -k\nabla\theta_h$ – вектор потока тепла; $h_f = 60$ (м⁻¹) – коэффициент теплообмена; $\theta_{ext} = 0^\circ\text{C}$ – температура окружающей среды. Все остальные границы будем считать теплоизолированными, т.е. такими, на которых нормальная компонента вектора потока тепла равна нулю: $\mathbf{n} \cdot \mathbf{q} = 0$.

Связанный термоупругий анализ. Предположим, что нижние грани буквы жестко закреплены и на нижнюю грань подается температура, а на верхней грани задано условие теплообмена, но механической нагрузки нет. Требуется рассчитать температурные напряжения.

РЕШЕНИЕ ЗАДАЧИ С ИСПОЛЬЗОВАНИЕМ ANSYS

Все необходимые параметры и материальные свойства определяется так же, как это было сделано ранее в случае плоских областей.

Здесь рассмотрим блок команд, создающих объемную фигуру из плоской области.

Построение объема с помощью вращения области вокруг заданной оси

Создадим объем путем вращения имеющейся области (с номером 3) на угол 45° вокруг оси, параллельной оси Oy и отстоящей от этой оси на расстояние WWL . Вращение области можно осуществить с помощью команды **VROTAT**. В результате работы этой команды будет создана новая сущность – объем (рис. 2).

! Определение точек оси, отстоящей от плоской области слева на WWL

K,1001,-WWL,0

K,1002,-WWL,HL

! Создать объем вращением области 3 вокруг оси с точками 1001 и 1002 на 45 градусов

VROTAT,3,,,,,1001,1002,45

Построение объема с помощью параллельного переноса области вдоль заданной линии

Рассмотрим случай построения трехмерной конструкции с помощью параллельного переноса (команда **VDRAG**) плоской области в форме буквы «R» вдоль оси z на $4*WWL$ м. Соответствующий блок команд будет выглядеть следующим образом:

```
! Определение точек и линий, определяющих траекторию переноса
K,1001,0,0
K,1002,0,0,4*WWL
! Новый стартовый номер для удобства нумерации дополнительной линии
NUMSTR,LINE,1001
L,1001,1002 ! линия с номером 1001
! Создать объем параллельным переносом области 3 вдоль линий
VDRAG,3,,,,,1001
```

Построение объема с помощью экструзии области (приращение координат и изменение масштаба)

Рассмотрим случай построения трехмерной конструкции с помощью экструзии (команда **VEXT**) плоской области в форме буквы «R» вдоль оси z на $4*WWL$ м с изменением масштаба - уменьшение в два раза. Соответствующий блок команд будет выглядеть следующим образом:

```
!ЭКСТРУЗИЯ -ПРИРАЩЕНИЕ КООРДИНАТ + МАСШТАБ
!VEXT, NA1, NA2, NINC, DX, DY, DZ, RX, RY, RZ
! Создать объем экструзией области 3
VEXT,3,,,,, 4*WWL,0.5,0.5
```

Построение конечно-элементной сетки


Рассмотрим блок команд, задающий параметры триангуляции и размер конечного элемента:

```
! Параметры для триангуляции
DMESH=WWL/4
! Установка общего размера конечного элемента
esize,dmesh
```

Также с помощью команды **KESIZE** можно задать сгущение сетки вокруг заданной точки, а с помощью команды **LESIZE** – размер конечного элемента либо число делений на заданной линии, например:

```
! разбиваем линию 1001 на DZ конечных элементов
LESIZE,1001,,,DZ
```

Зададим тип элемента и желаемый вид конечно-элементной сетки. Трехмерные элементы могут иметь различные формы (тетраэдр, призма, гексаэдр). На рисунке ниже показаны линейные элементы (без промежуточных узлов) и квадратичные элементы (с промежуточными узлами).



С помощью квадратичных тетраэдров (с промежуточными узлами) можно разбить на конечные элементы объем произвольной формы. Линейные тетраэдры (без промежуточных узлов) использовать не рекомендуется. Имеющиеся в ANSYS 11 элементы для структурного, температурного и термоупругого анализа собраны в таблицах 1-3.

Таблица 1. Элементы для структурного анализа (степени свободы: UX, UY в 2D; UX, UY, UZ в 3D)

Category	Element Name(s)
2-D Solids	PLANE25 , PLANE42 , PLANE82 , PLANE83 , VISCO88 , VISCO106 , VISCO108 , PLANE145 , PLANE146 , PLANE182 , PLANE183
3-D Solids	SOLID45 , SOLID46 , SOLID65 , VISCO89 , SOLID92 , SOLID95 , VISCO107 , SOLID147 , SOLID148 , SOLID185 , SOLID186 , SOLID187 , SOLID191 , SHELL281

Table 2.1 2-D температурные элементы (степени свободы: TEMP)

Element	Dimens.	Shape or Characteristic	DOFs
PLANE35	2-D	Triangle, 6-node	Temperature (at each node)
PLANE55	2-D	Quadrilateral, 4-node	Temperature (at each node)
PLANE75	2-D	Harmonic, 4-node	Temperature (at each node)
PLANE77	2-D	Quadrilateral, 8-node	Temperature (at each node)
PLANE78	2-D	Harmonic, 8-node	Temperature (at each node)

Table 2.2 3-D температурные элементы (степени свободы: TEMP)

Element	Dimens.	Shape or Characteristic	DOFs
SOLID70	3-D	Brick, 8-node	Temperature (at each node)
SOLID87	3-D	Tetrahedron, 10-node	Temperature (at each node)
SOLID90	3-D	Brick, 20-node	Temperature (at each node)

Таблица 3. Конечные элементы, используемые для структурно-температурного (термоупругого анализа)

Элементы	Моделируемые эффекты	Типы анализа
Линейные элементы SOLID5 - Coupled-Field Hexahedral PLANE13 - Coupled-Field Quadrilateral	Thermoelastic (Thermal Stress)	Static Full Transient
Квадратичные элементы SOLID98 - Coupled-Field Tetrahedral		
SOLID227 - Coupled-Field Tetrahedral PLANE223 - Coupled-Field Quadrilateral SOLID226 - Coupled-Field Hexahedral	Thermoelastic (Thermal Stress and Piezocaloric) Structural material nonlinearities Thermoplastic	Static Full Harmonic Full Transient

1) Каноническое разбиение трехмерной области гексаэдрами

В данном примере рассмотрим создание конечно-элементной сетки из гексаэдров на основе заданного шаблона разбиения области (каноническое разбиение).


Будем использовать восьмиузловые (линейные) гексаэдры. Тогда для проведения структурного анализа (решения задачи теории упругости) следует выбрать элемент SOLID45, а для проведения температурного анализа (решения задачи теплопроводности) – элемент SOLID70.

```
ET,1,SOLID70 ! 3-D 8-узловой элемент для температурного анализа, степени свободы: TEMP (температура)
! ET,1,SOLID45 ! ! 3-D 8-узловой элемент для структурного анализа, степени свободы: ux,uy,uz (перемещения)
```

Форма трехмерной фигуры позволяет построить разбиение гексаэдрами с помощью протягивания вспомогательной двумерной конечно-элементной сетки вдоль объема (получение гексаэдров из четырехугольников). Для этого используется вспомогательный геометрический конечный элемент, который не имеет степеней свободы, а служит только для задания геометрии сетки. Следует указать форму этого элемента: четырехузловой четырехугольник с помощью опции `keyopt(1)=6`. С помощью **MESH200** разбивается исходная плоская область.

```
ET,2,MESH200,6 ! FE of type 2: auxiliary mesh-only element, no degrees of freedom
```

Геометрия элемента MESH200



Для создания трехмерного конечно-элементного разбиения в результате протягивания полученной двумерной сетки вдоль объема используется команда **VSWEEP**.

`TYPE,2` ! Выбрать тип КЭ для разбиения плоской области (вспомогательный КЭ)

! Выбрать исходную двумерную область

`ASEL,S,LOC,Z,0`

! Разбить эту область вспомогательным элементом

`AMESH,ALL`

`TYPE,1` ! Выбрать тип КЭ для разбиения объемов (основной КЭ)

! Протянуть разбиение плоской области вдоль объема

`VSWEEP,ALL`

`Asel,all`

!`VSWEEP,ALL,<номер исходной области>,<номер конечной области>`

Конечно-элементная сетка для объема, полученного в результате вращения, показана на рис. 3, а для объема, полученного в результате параллельного переноса – на рис. 4 (используйте команду **EPLOT** или пункты меню Plot->Elements).




Рис. 3. Конечно-элементное разбиение гексаэдральными элементами: вращение




Рис. 4. Конечно-элементное разбиение гексаэдральными элементами: параллельный перенос

2) Разбиение трехмерной области тетраэдрами.

Объем произвольной формы всегда можно разбить на элементы тетраэдральной формы. В этом случае для разбиения объемов заданными трехмерными элементами используется стандартная команда генерации сетки **VMESH**, например:
ET,1,SOLID87 ! 3-D 10-узловой тетраэдральный элемент для температурного анализа, степени свободы: TEMP (температура)
VMESH,ALL ! Разбить все объемы на конечные элементы

Полученная конечно-элементная сетка для объема, полученного в результате вращения, показана на рис. 5.




Рис. 5. Конечно-элементное разбиение тетраэдральными элементами: вращение

Постановка граничных условий и решение задачи

Далее следует задать граничные условия и решить систему конечно-элементных уравнений (выполнив вход в решатель /**SOLU**).

Для примера рассмотрим задание граничных условий для температурного анализа для фигуры, полученной в результате вращения заданной области.

```
! Определение новой системы координат -
! цилиндрической с осью Z'', совпадающей с осью поворота
LOCAL,12,1,-WWL,0,0,, -90
CSYS,12
NSEL,S,LOC,Z,0 ! Выбор всех узлов с координатой Z''=0
NSEL,R,LOC,X,WWL,2*WWL
D,ALL,TEMP,T_INP1 ! Задать для всех выбранных узлов TEMP=T_INP1
NSEL,S,LOC,Z,0 ! Выбор всех узлов с координатой Z''=0
NSEL,R,LOC,X,WWL+WWL,WWL
D,ALL,TEMP,T_INP2 ! Задать для всех выбранных узлов TEMP=T_INP2
NSEL,S,LOC,Z,HL,HL ! Выбор всех узлов с координатой Z''=HL
NSEL,R,LOC,X,WWL,2*WWL
SF,ALL,CONV,H_F,T_EXT ! Конвективный теплообмен в выбранных узлах
NSEL,ALL ! Вернуться к выбору всех узлов модели
CSYS,0
```

Дадим комментарии к командам задания краевых условий. Анализ показывает, что в командном режиме получить доступ к нужным узлам лучше всего в цилиндрической системе координат $Ox''y''z''$, осью z'' которой является ось вращения исходной фигуры. Поэтому в команде **LOCAL** потребовалось сдвинуть начало координат новой системы на WWL влево по оси X , и повернуть систему координат относительно оси X на 90° в направлении вращения от оси Z к оси Y . (Согласно

справочной информации ANSYS это обеспечивается заданием угла вращения THYZ = -90° в соответствующем операнде команды LOCAL.) После перехода к цилиндрической системе координат под заявленным номером 12 по команде CSYS,12 команды выбора NSEL работает так, что ось X является осью R новой системы координат. Нужно также иметь ввиду, что новая ось Z в системе координат 12 соответствует бывшей оси Y плоской фигуры, сдвинутой влево на WWL.

Другой способ выбора необходимых узлов или элементов состоит в том, чтобы сначала выбрать области, содержащие нужные узлы, а затем использовать команду NSLA или ESLA (для элементов). Следующий блок команд задает условия конвективного теплообмена на области 9.

```
ASEL,S,AREA,,9
NSLA,S ! Выбрать узлы, принадлежащие выбранным областям
! Конвективный теплообмен в выбранных узлах
SF,ALL,CONV,H_F,T_EXT
! выбор всего
ALLSEL
```

Температура в выбранных узлах задается с помощью команды D. Конвективный теплообмен в выбранных узлах задается с помощью команды SF. Применение данных команд для трехмерного случая аналогично их применению в двумерном случае.

Для структурного анализа перемещения в выбранных узлах задаются с помощью команды D. Распределенная нагрузка в выбранных узлах задается с помощью команды SF. Применение данных команд для трехмерного случая также аналогично их применению в двумерном случае.

Для задания граничных условий на выбранных областях можно использовать, соответственно, команды DA и SFA как для температурного, так и для структурного анализа.

В случае термоупругого анализа в описание материальных свойств нужно добавить плотность и коэффициент температурного расширения.

```
!-----
! "Температурные" материальные свойства
T_INP1=90 ! Температура на левой части нижней границе
T_INP2=180 ! Температура на правой части нижней границе
T_EXT=0 ! Температура окружающей среды для конвективных условий
T_EXT=0
H_F=60 ! Коэффициент конвективного теплообмена

MP,KXX,1,46.7 ! Коэффициент теплопроводности KXX=46.7 (Сталь)

ALPH1=1.51e-05 !коэффициент температурного расширения
MP,ALPX,1,ALPH1
!-----
! "Упругие" материальные свойства
! Материальные константы стали
RO1=7.8e3 ! плотность
E1=2.1e11 !модуль Юнга
NU1=0.29 !коэффициент Пуассона
```

```

MP, DENS, 1, RO1
MP, EX, 1, E1
MP, NUXY, 1, NU1
!-----
Все механические и температурные условия обычным образом:
! Решение связанной задачи термоупругости
/SOLU
ANTYPE, STAT          ! Решение стационарной задачи
! Условия для "температурного" анализа
TREF, 0 ! Определение относительной температуры (в градусах)
! Определение новой системы координат -
! цилиндрической с осью Z'', совпадающей с осью поворота
LOCAL, 12, 1, -WWL, 0, 0, , -90
CSYS, 12
NSEL, S, LOC, Z, 0    ! Выбор всех узлов с координатой Z''=0
NSEL, R, LOC, X, WWL, 2*WWL
D, ALL, TEMP, T_INP1    ! Задать для всех выбранных узлов TEMP=T_INP1
NSEL, S, LOC, Z, 0    ! Выбор всех узлов с координатой Z''=0
NSEL, R, LOC, X, WL+WWL, WL
D, ALL, TEMP, T_INP2    ! Задать для всех выбранных узлов TEMP=T_INP2
NSEL, S, LOC, Z, HL, HL ! Выбор всех узлов с координатой Z''=HL
NSEL, R, LOC, X, WWL, 2*WWL
SF, ALL, CONV, H_F, T_EXT ! Конвективный теплообмен в выбранных узлах
NSEL, ALL              ! Вернуться к выбору всех узлов модели
CSYS, 0


! Условия для "упругого" анализа
ASEL, S, AREA, , 1, 5, 4
NSLA, S, 1
D, ALL, UX, 0
D, ALL, UY, 0
D, ALL, UZ, 0
ASEL, ALL
NSEL, ALL              ! Вернуться к выбору всех узлов модели
SOLVE
FINISH

```

АНАЛИЗ РЕЗУЛЬТАТОВ для термоупругого анализа

Граничные условия для термоупругого анализа показаны на конечно-элементной модели для трехмерной фигуры, полученной из плоской твердотельной модели путем вращения плоской области вокруг оси (рис. 6). (Пункт меню: Plot->Elements, для показа граничных условий PltCtrls->Symbols->tick All applied BC, условия конвективного теплообмена: Surface Load Symbols->Convective FilmCoef):

ELEMENTS
U
TEMP
CONV-HCOE
60




Thermal stress in 3D region (R)

Рис. 6 Конечно-элементная модель с граничными условиями

Деформированная форма конструкции представлена на рис. 7.

DISPLACEMENT
STEP=1
SUB =1
TIME=1
DMX =.004448



Thermal stress in 3D region (R)

Рис. 7 Деформированная конечно-элементная сетка и недеформированный контур

Результат вычисления полей температуры и вектора потока тепла представлены на рис. 8 и 9.




Рис. 8 Распределение температуры в трехмерной конструкции, полученной методом вращения




Рис. 9 Распределение вектора потока тепла

Рис. 10 иллюстрирует распределение модуля вектора перемещений.




Рис. 10 Распределение модуля вектора перемещений

Рис. 11 иллюстрирует распределение интенсивности напряжений.




Рис. 11 Распределение интенсивности напряжений